
An Erlang Binding to FoundationDB
    

[image: CI]
An Erlang library that wraps the FoundationDB C API with a NIF.
We also provide a conforming implementation of the Tuple and Directory layers.
This project originated as apache/couchdb-erlfdb. This fork
was created in early 2024 to continue development.

  
    
  
  Dependencies


Refer to FoundationDB's Getting Started on Linux or Getting Started on macOS.
Install the foundationdb-clients package to run erlfdb as a client to an existing
FoundationDB database.
Install the foundationdb-server package for running a test fdbserver on your local device.
This package is required to run the unit tests.

  
    
  
  Usage



  
    
  
  Erlang's rebar3


Add erlfdb as a dependency in your Erlang project's rebar.config:
% rebar.config
{deps, [
    {erlfdb, "0.3.0"}
]}.

  
    
  
  Elixir's Mix


Add erlfdb as a dependency in your Elixir project's mix.exs:
# mix.exs
defp deps do
  [
    {:erlfdb, "~> 0.3"}
  ]
end

  
    
  
  Example


A simple example showing how to open a database and read and write keys.
See the erlfdb Documentation for more.

  
    
  
  Erlang


1> Db = erlfdb:open(<<"/usr/local/etc/foundationdb/fdb.cluster">>).
{erlfdb_database,#Ref<0.2859661758.3941466120.85406>}
2> ok = erlfdb:set(Db, <<"foo">>, <<"bar">>).
ok
3> erlfdb:get(Db, <<"foo">>).
<<"bar">>
4> erlfdb:get(Db, <<"bar">>).
not_found

  
    
  
  Elixir


iex> db = :erlfdb.open("/usr/local/etc/foundationdb/fdb.cluster")
{:erlfdb_database, #Reference<0.2859661758.3941466120.85406>}
iex> :ok = :erlfdb.set(db, "foo", "bar")
:ok
iex> :erlfdb.get(db, "foo")
"bar"
iex> :erlfdb.get(db, "bar")
:not_found

  
    
  
  Binding Tester


FoundationDB has a custom binding tester that can be used to test whether
changes have broken compatibility. The GitHub Action runs the Binding Tester
against the most recent supported version of FoundationDB.

  
    
  
  Developing erlfdb



  
    
  
  Building


$ rebar3 compile


  
    
  
  Testing


# rebar3 eunit

  
    
  
  Bypassing dependency checks


When you execute a rebar command, erlfdb attempts to detect the version of the fdbcli
installed on your system. If it cannot be detected, the rebar command is aborted.
To disable the abort, use ERLFDB_ASSERT_FDBCLI=0. For example, you can safely use
this for the fmt command, which does not do a compile action.
ERLFDB_ASSERT_FDBCLI=0 rebar3 fmt




  

    
LICENSE
    



                                Apache License
                          Version 2.0, January 2004
                       http://www.apache.org/licenses/

  TERMS AND CONDITIONS FOR USE, REPRODUCTION, AND DISTRIBUTION

  1. Definitions.

     "License" shall mean the terms and conditions for use, reproduction,
     and distribution as defined by Sections 1 through 9 of this document.

     "Licensor" shall mean the copyright owner or entity authorized by
     the copyright owner that is granting the License.

     "Legal Entity" shall mean the union of the acting entity and all
     other entities that control, are controlled by, or are under common
     control with that entity. For the purposes of this definition,
     "control" means (i) the power, direct or indirect, to cause the
     direction or management of such entity, whether by contract or
     otherwise, or (ii) ownership of fifty percent (50%) or more of the
     outstanding shares, or (iii) beneficial ownership of such entity.

     "You" (or "Your") shall mean an individual or Legal Entity
     exercising permissions granted by this License.

     "Source" form shall mean the preferred form for making modifications,
     including but not limited to software source code, documentation
     source, and configuration files.

     "Object" form shall mean any form resulting from mechanical
     transformation or translation of a Source form, including but
     not limited to compiled object code, generated documentation,
     and conversions to other media types.

     "Work" shall mean the work of authorship, whether in Source or
     Object form, made available under the License, as indicated by a
     copyright notice that is included in or attached to the work
     (an example is provided in the Appendix below).

     "Derivative Works" shall mean any work, whether in Source or Object
     form, that is based on (or derived from) the Work and for which the
     editorial revisions, annotations, elaborations, or other modifications
     represent, as a whole, an original work of authorship. For the purposes
     of this License, Derivative Works shall not include works that remain
     separable from, or merely link (or bind by name) to the interfaces of,
     the Work and Derivative Works thereof.

     "Contribution" shall mean any work of authorship, including
     the original version of the Work and any modifications or additions
     to that Work or Derivative Works thereof, that is intentionally
     submitted to Licensor for inclusion in the Work by the copyright owner
     or by an individual or Legal Entity authorized to submit on behalf of
     the copyright owner. For the purposes of this definition, "submitted"
     means any form of electronic, verbal, or written communication sent
     to the Licensor or its representatives, including but not limited to
     communication on electronic mailing lists, source code control systems,
     and issue tracking systems that are managed by, or on behalf of, the
     Licensor for the purpose of discussing and improving the Work, but
     excluding communication that is conspicuously marked or otherwise
     designated in writing by the copyright owner as "Not a Contribution."

     "Contributor" shall mean Licensor and any individual or Legal Entity
     on behalf of whom a Contribution has been received by Licensor and
     subsequently incorporated within the Work.

  2. Grant of Copyright License. Subject to the terms and conditions of
     this License, each Contributor hereby grants to You a perpetual,
     worldwide, non-exclusive, no-charge, royalty-free, irrevocable
     copyright license to reproduce, prepare Derivative Works of,
     publicly display, publicly perform, sublicense, and distribute the
     Work and such Derivative Works in Source or Object form.

  3. Grant of Patent License. Subject to the terms and conditions of
     this License, each Contributor hereby grants to You a perpetual,
     worldwide, non-exclusive, no-charge, royalty-free, irrevocable
     (except as stated in this section) patent license to make, have made,
     use, offer to sell, sell, import, and otherwise transfer the Work,
     where such license applies only to those patent claims licensable
     by such Contributor that are necessarily infringed by their
     Contribution(s) alone or by combination of their Contribution(s)
     with the Work to which such Contribution(s) was submitted. If You
     institute patent litigation against any entity (including a
     cross-claim or counterclaim in a lawsuit) alleging that the Work
     or a Contribution incorporated within the Work constitutes direct
     or contributory patent infringement, then any patent licenses
     granted to You under this License for that Work shall terminate
     as of the date such litigation is filed.

  4. Redistribution. You may reproduce and distribute copies of the
     Work or Derivative Works thereof in any medium, with or without
     modifications, and in Source or Object form, provided that You
     meet the following conditions:

     (a) You must give any other recipients of the Work or
         Derivative Works a copy of this License; and

     (b) You must cause any modified files to carry prominent notices
         stating that You changed the files; and

     (c) You must retain, in the Source form of any Derivative Works
         that You distribute, all copyright, patent, trademark, and
         attribution notices from the Source form of the Work,
         excluding those notices that do not pertain to any part of
         the Derivative Works; and

     (d) If the Work includes a "NOTICE" text file as part of its
         distribution, then any Derivative Works that You distribute must
         include a readable copy of the attribution notices contained
         within such NOTICE file, excluding those notices that do not
         pertain to any part of the Derivative Works, in at least one
         of the following places: within a NOTICE text file distributed
         as part of the Derivative Works; within the Source form or
         documentation, if provided along with the Derivative Works; or,
         within a display generated by the Derivative Works, if and
         wherever such third-party notices normally appear. The contents
         of the NOTICE file are for informational purposes only and
         do not modify the License. You may add Your own attribution
         notices within Derivative Works that You distribute, alongside
         or as an addendum to the NOTICE text from the Work, provided
         that such additional attribution notices cannot be construed
         as modifying the License.

     You may add Your own copyright statement to Your modifications and
     may provide additional or different license terms and conditions
     for use, reproduction, or distribution of Your modifications, or
     for any such Derivative Works as a whole, provided Your use,
     reproduction, and distribution of the Work otherwise complies with
     the conditions stated in this License.

  5. Submission of Contributions. Unless You explicitly state otherwise,
     any Contribution intentionally submitted for inclusion in the Work
     by You to the Licensor shall be under the terms and conditions of
     this License, without any additional terms or conditions.
     Notwithstanding the above, nothing herein shall supersede or modify
     the terms of any separate license agreement you may have executed
     with Licensor regarding such Contributions.

  6. Trademarks. This License does not grant permission to use the trade
     names, trademarks, service marks, or product names of the Licensor,
     except as required for reasonable and customary use in describing the
     origin of the Work and reproducing the content of the NOTICE file.

  7. Disclaimer of Warranty. Unless required by applicable law or
     agreed to in writing, Licensor provides the Work (and each
     Contributor provides its Contributions) on an "AS IS" BASIS,
     WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or
     implied, including, without limitation, any warranties or conditions
     of TITLE, NON-INFRINGEMENT, MERCHANTABILITY, or FITNESS FOR A
     PARTICULAR PURPOSE. You are solely responsible for determining the
     appropriateness of using or redistributing the Work and assume any
     risks associated with Your exercise of permissions under this License.

  8. Limitation of Liability. In no event and under no legal theory,
     whether in tort (including negligence), contract, or otherwise,
     unless required by applicable law (such as deliberate and grossly
     negligent acts) or agreed to in writing, shall any Contributor be
     liable to You for damages, including any direct, indirect, special,
     incidental, or consequential damages of any character arising as a
     result of this License or out of the use or inability to use the
     Work (including but not limited to damages for loss of goodwill,
     work stoppage, computer failure or malfunction, or any and all
     other commercial damages or losses), even if such Contributor
     has been advised of the possibility of such damages.

  9. Accepting Warranty or Additional Liability. While redistributing
     the Work or Derivative Works thereof, You may choose to offer,
     and charge a fee for, acceptance of support, warranty, indemnity,
     or other liability obligations and/or rights consistent with this
     License. However, in accepting such obligations, You may act only
     on Your own behalf and on Your sole responsibility, not on behalf
     of any other Contributor, and only if You agree to indemnify,
     defend, and hold each Contributor harmless for any liability
     incurred by, or claims asserted against, such Contributor by reason
     of your accepting any such warranty or additional liability.

  END OF TERMS AND CONDITIONS

  APPENDIX: How to apply the Apache License to your work.

     To apply the Apache License to your work, attach the following
     boilerplate notice, with the fields enclosed by brackets "[]"
     replaced with your own identifying information. (Don't include
     the brackets!)  The text should be enclosed in the appropriate
     comment syntax for the file format. We also recommend that a
     file or class name and description of purpose be included on the
     same "printed page" as the copyright notice for easier
     identification within third-party archives.

  Copyright [yyyy] [name of copyright owner]

  Licensed under the Apache License, Version 2.0 (the "License");
  you may not use this file except in compliance with the License.
  You may obtain a copy of the License at

      http://www.apache.org/licenses/LICENSE-2.0

  Unless required by applicable law or agreed to in writing, software
  distributed under the License is distributed on an "AS IS" BASIS,
  WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
  See the License for the specific language governing permissions and
  limitations under the License.




  

    
Changelog
    


  
    
  
  v0.3 (2025-04-03)



  
    
  
  Enhancements


	(#46) Add support for client_threads_per_version config. See FDB Client Threads for details.


  
    
  
  New contributors


Thank you to the following new contributors! :)
	weaversam8


  
    
  
  v0.2.4 (2025-03-23)



  
    
  
  Bug fixes


	(#43) For OTP >= 26, erlfdb NIF will no longer segfault on VM halt.


  
    
  
  v0.2.3 (2025-03-15)



  
    
  
  Bug fixes


	(#36) Starting in v0.2.2, the ready message from a get_versionstamp future included
the ref from the transaction, putting it at risk of being flushed in the after the
transaction completes. Now, the ready message behaves much like the watch, where it
is not tied to the transaction ref.
	Sandboxes can now be specified with a subdirectory, allowing concurrent
creation (erlfdb_sandbox)


  
    
  
  Enhancements


	Livebook | KvQueue: An interactive notebook that
demonstrates the creation of a durable queue in FoundationDB.



  
    
  
  v0.2.2 (2024-01-16)



  
    
  
  Bug fixes


	(#31) Previously, erlfdb could leak {reference(), ready} messages to the caller if
the transaction UserFun was executed more than once. We will now flush such messages before
transactional/2 returns control to the caller. Watches are unaffected.
	(#32) Fixed documentation for versionstamps in erlfdb_tuple.


  
    
  
  Enhancements


	erlfdb_sandbox: Creates and starts a database on your local filesystem that is to be
used as a sandbox. That is, it can be useful for development tasks like tutorials and unit tests.
It should not be used in a production setting.


  
    
  
  New contributors


Thank you to the following new contributors! :)
	drowzy


  
    
  
  v0.2.1 (2024-11-20)



  
    
  
  Testing


	The FDB Bindings Tester now runs under GitHub Actions
	Added macOS GitHub Action


  
    
  
  v0.2.0 (2024-09-21)



  
    
  
  Bug fixes


	Several type specs in erlfdb were corrected.


  
    
  
  Enhancements


	erlfdb:wait_for_all_interleaving/2: Given a list of fold_future() or future(), calling this function will wait
on the futures at once, and then continue to issue any remaining get_range or get_mapped_range until
the result set is exhausted. This allows fdbserver to process multiple get_ranges at the same time,
in a pipelined fashion.
	erlfdb:get_range_split_points/4: Companion to wait_for_all_interleaving, this is an fdbserver-supported function
that will split a given key range into a partitioning set of ranges for which the key-values for each section are
approximately equal to the provided chunk_size option. There are limitations to this, namely that a hard
maximum of 100 shards can be traversed. The default chunk_size is 10000000 (in Bytes).
	erlfdb:get_range*: The default behavior of get_range is now more explicit in the type specs and with the wait
option, with defaults to true. A value of false will return a fold_future(), and interleaving is
an experimental feature that will use both get_range_split_points and wait_for_all_interleaving to retrieve the range.




  

    
FDB Client Threads
    

The default configuration of erlfdb includes the creation and management of a single OS thread that is responsible for all database work created by processes in the Erlang VM. As your workload increases in size, you will need to configure the erlfdb client for horizontal scaling. Otherwise, a sufficiently sized system with a sufficiently sized workload can saturate the single client thread with work, which will lead to high latency on individual requests, and a ceiling on throughput.
This page will guide you through the configuration necessary to scale the FDB client to several worker threads on a single BEAM VM.

  
    
  
  Network options configuration


The configuration will be done via the erlfdb network_options environment variable. In erlfdb >= 0.3, defaults are chosen that are meant to ease the configuration burden. The options are described in detail below, but the main takeaway is the following:
Tip
In order to double your client's maximum throughput, set
client_threads_per_version to 2.

[
    {erlfdb, [
        {network_options, [
            {client_threads_per_version, 2}
        ]}
    ]}
].
Please read on for details about the scaling implementation.

  
    
  
  Version 0.3 and above


In versions 0.3 and above, the env var network_options defaults to
[
    {erlfdb, [
        {network_options, [
            % `ignore_external_client_failures`:
            % Optional: Allows your client to prepare fdbserver version
            % upgrades. When a future client library fails to load, it's not a
            % failure condition, because we know it will be valid at some
            % point in the future.
            {ignore_external_client_failures, true},

            % `callbacks_on_external_threads`:
            % Allows libfdb_c to apply future callbacks on external threads,
            % which increases the throughput of future resolution. erlfdb
            % simply sends the result to the required Erlang process, so the
            % default behavior of executing on the main thread is not
            % desirable
            {callbacks_on_external_threads, true},

            % `external_client_directory`:
            % Points libfdb_c to the directory containing the dynamic
            % library(ies) to be loaded. libfdb_c will copy each .so N times
            % to temporary files on the filesystem so that they can be
            % individually loaded. Normally the value is a binary string of
            % the directory on the filesystem containing the dynamic
            % library(ies).
            %
            % You may also provide an `{M, F, A}` tuple. If you do, the function
            % is executed at the time the `erlfdb_nif` module is loaded. The
            % return must be a binary string. This value is used as the
            % permanent value for `external_client_directory` for the lifetime
            % of the VM.
            {external_client_directory, {erlfdb_network_options, compile_time_external_client_directory, []}},

            % `client_threads_per_version`:
            % Number of threads to create per dynamic library. This is the
            % value to tweak in order to scale the number of client threads
            % horizontally. This value is used as the permanent value
            % for the lifetime of the BEAM VM.
            %
            % When the value is 1, only the local client is used. When the
            % value is N > 1, then N external client threads are used.
            %
            % You may also provide an `{M, F, A}` tuple. If you do, the
            % function is executed at the time the `erlfdb_nif` module is
            % loaded. The return must be a positive integer.
            %
            % Note: For each thread, libfdb_c will create a tcp
            % connection for each coordinator in the cluster file.
            {client_threads_per_version, 1}
        ]}
    ]}
].
When the erlfdb NIF is loaded, any network_options defined by your config are merged into the defaults, so that yours always take precedence. The final set of resolved network_options is stored in env var network_options_resolved.
application:get_env(erlfdb, network_options_resolved).

  
    
  
  How to choose the correct value for client_threads_per_version


The ideal approach is to choose an integer value that is sufficient for your workload, and no more. It's not recommended to choose a value that is larger than the number of online schedulers.
Instead of choosing a specific value, you may wish for erlfdb to scale dynamically along with your scaling of the Erlang VM itself. Consider the following MFA-tuple choices.
	{erlang, system_info, [schedulers_online]}
	{erlang, system_info, [dirty_io_schedulers]}

Info
These suggestions are made simply as a convenience for automatic scaling, and none of these choices will associate erlfdb external client threads with Erlang's schdeulers or dirty_io_schedulers. External client threads will always be created in addition to and not replacements for any of the Erlang VM's OS threads.


  
    
  
  Making use of External Client Threads



  
    
  
  Distributing your workfload


So you've configured the client to create multiple external client threads. Now, you need to make sure your application distributes the workload among these threads.
The function erlfdb:open/2 provides your application with a convenient way to distribute your workload. Whenever you need a database object from which to create a transaction, call open to retrieve one very quickly (via persistent_term storage). With the default work distribution strategy (scheduler_id), erlfdb will create and keep track of N database objects where N is the number of schedulers on your system. Each of these database objects will automatically be assigned to one of the external client threads in libfdb_c.
Info
This strategy does not claim that a particular item of work will remain on an Erlang scheduler -- only that the scheduler_id itself is helpful in distributing work in a roughly even fashion.

It may be beneficial to create all database objects during startup. The function erlfdb:open_all/2 is provided for this purpose.

  
    
  
  Thread lifecycle


When client_threads_per_version == 1,
	enif_thread_create is called exactly once. This is the 'local client network thread' a.k.a. the libfdb_c Main Thread. This thread is given the name fdb:network_thread.
	Shutdown: The erlfdb NIF uses enif callbacks to ensure the event loop is stopped and the thread is joined before the VM is terminated.

When client_threads_per_version > 1, the behavior described above is true; additionally:
	libfdb_c creates N copies of the dynamic library into temp files on the filesystem. Each copy will house 1 of the external threads.
	N threads are created by libfdb_c. These threads are not created with enif_thread_create, because the creation is contained in logic internal to libfdb_c. Each of these threads is given the name fdb-<vsn>-<index>. If the thread name is longer than 15 chars, it's instead given the name fdb-<vsn>. If this is longer than 15 chars, it's given the name fdb-external. On a Linux system, these threads are visible with top -H -p $beam_pid.
	Each database object (via erlfdb:create_database/1) is linked to a client thread at the time of creation. The threads are distributed in a round-robin fashion. Therefore, to make use of N client threads, you must have N database objects.
	Shutdown: Each external thread is waited upon immediately after the local client network event loop returns. Thus, you may consider the external client threads as "children" of the local client network thread created by enif_thread_create. This relationship is necessary and sufficient for the Erlang VM and its operator to maintain control over the OS threads on the system.


  
    
  
  Tracing


To enable client tracing (C API Option FDB_NET_OPTION_TRACE_ENABLE), set the following:
[
    {erlfdb, [
        {network_options, [
            % `trace_enable`:
            % A directory on the filesystem where trace files will be written.
            {trace_enable, <<"/path/to/existing/directory">>},

            % `trace_format`:
            % The format of the trace files. Supported formats are `<<"xml">>` (the default) and `<<"json">>`.
            {trace_format, <<"xml">>}
        ]}
    ]}
].

  
    
  
  Past Versions 0.0 - 0.2


In versions 0.0.x - 0.2.x, the env var network_options defaults to []. The behavior of the erlfdb NIF with respect to thread creation is equivalent to having client_threads_per_version == 1, as decribed above.
In these erlfdb versions, the client_threads_per_version env var is not supported, so horizontal scaling of a single client is not possible. Instead, consider starting multiple Erlang VMs in order to distribute your workload, or upgrade to erlfdb >= 0.3.

  
    
  
  References


A short list of suggested reading in the FoundationDB source that will help the reader understand the details of the client threading model.
	fdbclient/MultiVersionTransaction.actor.cpp: implements the management of the local and external clients.
	bindings/c/fdb_c.cpp: defines MultiVersionApi as the default implementation
	fdbclient/NativeAPI.actor.cpp: implements the client (local or external). Specifically, maintains the reference to the thread.
	flow/Net2.actor.cpp: implements the event loop




  

    
KvQueue - A distributed durable queue with erlfdb
    

Mix.install([
  {:erlfdb, "~> 0.3"}
])

  
    
  
  Introduction


The FoundationDB Layer Concept means that our query execution is detached from our storage servers, allowing a database operator to dynamically scale compute nodes according to current query workloads. However, there is also an advantage for application developers: when combined with transactions, a Layer allows us to create powerful data structures as building blocks for an application.
In this tutorial, we'll create a Layer that is a performant durable queue. Then, we'll consume items from this queue using a gen_server.
Our queue will prioritize throughput of the producer. This is a common design choice for a durable queue because it ensures our producer can quickly move on to other operations in our application, without costly locks or conflicts to slow it down.
With transactional isolation guarantees from FDB, we can create as many parallel and distributed producers and consumers as we like; a given message will always be consumed exactly once.

  
    
  
  Start a sandbox


First, we'll create a sandbox database that we can freely write to and delete from. erlfdb_sandbox starts a single fdbserver process and writes data to a temporary directory. It may take a few seconds the first time you execute it.
Db = erlfdb_sandbox:open().

  
    
  
  Defining the Queue


We'll define a standard FIFO queue with push and pop operations. Our main challenge is laying out the queue onto FDB key-value pairs. It's up to us to design a keyspace that will fit our design specifications. So, let's discuss it now.

  
    
  
  Key-value binary format


Our keyspace will use the FDB Tuple Layer for constructing the key format. We'll document our keys and values with this simple notation:
{<some-tuple-key>} => <some-binary-value>

  
    
  
  Queue name


We're ready to talk keys! First, we'll hard-code the name of our queue to <<"q">>. This makes our queue a singleton in the database, which keeps our code concise for this tutorial. The data structures you create should allow a developer to specify certain details like the name to avoid key conflicts.
All our key-values will be of the form:
{q, ...} => <some-value>

  
    
  
  Queue length


Next, in order to keep track of the length of the queue, we'll keep track of counter values called <<"npush">> and <<"npop">>, which are the total number of pushes and pops executed,
respectively. Splitting the counters is an important part of how the consumer works, which we'll get to later.
{q, npush} => <push-count>
{q, npop} => <pop-count>
At any given time, the length of the queue is equal to <push-count> - <pop-count>. We'll have to make sure these counters are safely managed so that there are no race conditions. (We don't want negative values!)

  
    
  
  Queue items


Finally, keys for items in the queue include <<"val">> in the key:
{q, val, <vs>} => <item>
where <vs> is a placeholder for a versionstamp, and <item> is the content of item stored in the queue.
A Versionstamp is a unique monotinically increasing integer generated by FDB at commit-time. We have to use special client-side functions that operate on "incomplete versionstamps" because the client cannot know what the value of the versionstamp is before the commit.
Using a versionstamp for pushing items onto the queue ensures that there is no key contention for parallel pushes, which greatly increases producer throughput.

  
    
  
  Code time!


Once you have a good grasp of the keyspace, writing the code becomes pretty straightforward!
-module(kv_queue).
-export([
         delete/1,
         push/2,
         consume_k/2,
         pop/1,
         pop_k/2,
         len/1,
         queue/1,
         watch/1
        ]).

-define(NAME, <<"q">>).
-define(VAL, <<"val">>).
-define(NPOP, <<"npop">>).
-define(NPUSH, <<"npush">>).

% This is known as an "incomplete versionstamp"
-define(VS(Tx), {
      versionstamp,
      16#ffffffffffffffff,
      16#ffff,
      erlfdb:get_next_tx_id(Tx)
    }).

delete(Tx) ->
  {S, E} = erlfdb_tuple:range({?NAME}),
  erlfdb:clear_range(Tx, S, E).

push(Tx, Val) ->
  Key = erlfdb_tuple:pack_vs({?NAME, ?VAL, ?VS(Tx)}),
  erlfdb:set_versionstamped_key(Tx, Key, Val),
  erlfdb:add(Tx, erlfdb_tuple:pack({?NAME, ?NPUSH}), 1).

pop(Tx) ->
  case pop_k(Tx, 1) of
    {_, [Val]} -> {ok, Val};
    {_, []} -> error
  end.

pop_k(Tx, K) ->
  {QS, QE} = erlfdb_tuple:range({?NAME, ?VAL}),
  case erlfdb:get_range(Tx, QS, QE, [{limit, K}, {wait, true}]) of
    [] ->
      {{error, empty}, []};
    KVs=[{S, _}|_] ->
      N = length(KVs),
      {E, _} = lists:last(KVs),
      erlfdb:clear_range(Tx, S, erlfdb_key:strinc(E)),
      erlfdb:add(Tx, erlfdb_tuple:pack({?NAME, ?NPOP}), N),

      Status = if N == K -> ok; true -> {error, empty} end,
      {Status, [ V || {_, V} <- KVs ]}
  end.

len(Tx) ->
  F = [
    erlfdb:get(Tx, erlfdb_tuple:pack({?NAME, ?NPUSH})),
    erlfdb:get(Tx, erlfdb_tuple:pack({?NAME, ?NPOP}))
  ],
  [Npush, Npop] = [ decode_as_int(X, 0) || X <- erlfdb:wait_for_all(F)],
  Npush - Npop.

queue(Tx) ->
  {QS, QE} = erlfdb_tuple:range({?NAME, ?VAL}),
  KVs = erlfdb:get_range(Tx, QS, QE, [{wait, true}]),
  [ V || {_, V} <- KVs ].

decode_as_int(not_found, Default) -> Default;
decode_as_int(Val, _Default) -> binary:decode_unsigned(Val, little).

consume_k(Tx, K) ->
  case pop_k(Tx, K) of
    {ok, Vals} ->
      {Vals, undefined};
    {{error, empty}, Vals} ->
      {Vals, watch(Tx)}
  end.

watch(Tx) ->
  erlfdb:watch(Tx, erlfdb_tuple:pack({?NAME, ?NPUSH})).

Functions push/2 and pop_k/2 are the two most important functions to focus on for now.
We'll discuss consume_k/2, and watch/1 later, when we define our consumer.

  
    
  
  Queue push


When we push an item onto the queue, we want to make sure that there is no key contention. So,
	We insert the item into the val keyspace with an incomplete versionstamp in the key. This ensures that n parallel transactions will get n unique versionstamps and insert into the queue in the order of the serialized commits, with no key conflicts.
	We do an atomic add on the npush key. FDB atomic operations are CRDTs. That is, FDB can resolve parallel atomic operations without key conflicts. An interesting consequence of this is that our client does not know the value of the counter when incrementing it.


  
    
  
  Queue pop


When we pop k items from the front of the queue, we want to make sure we safely delete them in the same transaction.
	A get_range with limit will read up to k items in the range and then stop.
	When values are found, we use clear_range carefully to only touch those we've retrieved, and increment the npop counter accordingly.
	Smart return values will help our consumer know when to keep consuming and when to wait.


  
    
  
  Try it out!


We can push any item represented by a binary() onto the queue. If you execute this multiple times, a <<"hello">> item will be added each time.
erlfdb:transactional(Db, fun(Tx) -> kv_queue:push(Tx, <<"hello">>) end).
We can view the entire queue at once. This is not how we will be consuming items, but it's nice to see them all at the same time.
erlfdb:transactional(Db, fun kv_queue:queue/1).
We can pop a single item from the front of the queue. If you execute this until the queue is empty, you'll receive error as the response. That just means that the queue is empty.
erlfdb:transactional(Db, fun kv_queue:pop/1).
If you've emptied the queue, the length will be 0. Feel free to go back and add some more items.
erlfdb:transactional(Db, fun kv_queue:len/1).

  
    
  
  Defining the Consumer


Our consumer is a gen_server. We'll start any number of them, and each one will monitor the queue for items. There is no coordination between consumers. They will each try to empty the queue as quickly as possible, and then wait for a signal to start once again.
The consumer will be configurable to consume K items from a single transaction. Increasing K increases consumer throughput quite dramatically.
Note: consumers will have lots of key conflicts with each other because they are all vying for the items at the front of the queue. Other clever optimizations can be done to reduce conflicts, but we're skipping that for now.
-module(kv_consumer).
-export([
         start_link/1,
         init/1,
         handle_cast/2,
         handle_call/3,
         handle_info/2
]).

-behaviour(gen_server).

start_link(InitArgs) ->
  gen_server:start_link(?MODULE, InitArgs, []).

init([Db, K]) ->
  gen_server:cast(self(), consume),
  {ok, #{db => Db, k => K, watch => undefined}}.

handle_cast(consume, S) ->
  #{db := Db, k := K} = S,
  {Items, Watch} =
          erlfdb:transactional(Db,
                  fun(Tx) -> kv_queue:consume_k(Tx, K) end
          ),
  [ log("Consumed items ~p~n", [Items]) || Items =/= [] ],
  case Watch of
    undefined ->
      gen_server:cast(self(), consume);
    _ ->
      log("Waiting for push~n", [])
  end,
  {noreply, S#{watch => Watch}}.

handle_call(_Call, _From, _S) ->
  erlang:error(function_clause).

handle_info({Ref, ready}, S=#{watch := {erlfdb_future, Ref, _}}) ->
  gen_server:cast(self(), consume),
  {noreply, S#{watch => undefined}};
handle_info(_Info, S) ->
  {noreply, S}.

log(F, A) ->
  io:format("~p: " ++ F, [self()|A]).

  
    
  
  Watch for pushes


Up above we skipped over the watch/2 function. Here it shines. Via consume_k/2, when the queue is empty, the consumer receives a "watch" on the npush counter. Whenever the npush key's value is changed, FDB resolves all watches on that key: each process that created a watch receives a {reference(), ready} message on its message queue.
Remember earlier we decided to track the queue length with two counters instead of one? This is why: we don't want our consumers to be signaled by other consumers; we only want to know when new items are available. By splitting the queue length into npush and npop, we allow watch creation only on npush.
The consumer uses the watch to enter an idle state while the queue is empty, so we don't have to continuously poll for the existence of queue items.
In summary, consume_k/2 will attempt to consume K items from the front of the queue. If any number less than K is found, then a watch is created, and the consumer goes into an idle state.

  
    
  
  Processing queue items


It's critically important to do all processing of queue items outside of the FDB transaction. For our simple case, we're just logging the items, but notice that our log is not within the erlfdb:transactional/2 function. Because a transaction function can execute multiple times for a single commit (e.g. when there are key conflicts), side-effects inside transactions are quite dangerous.

  
    
  
  Supervisor


This is a standard simple_one_for_one supervisor. Nothing special about it. We're defining it here to help us keep track of all the consumers we're about to start.
-module(kv_consumer_sup).
-behaviour(supervisor).

-export([start_link/0, start_child/2, terminate_all_children/0, init/1]).

start_link() -> supervisor:start_link({local, ?MODULE}, ?MODULE, []).

start_child(Db, K) ->
  supervisor:start_child(?MODULE, [[Db, K]]).

terminate_all_children() ->
  [ supervisor:terminate_child(?MODULE, X)
   || {_, X, _, _} <- supervisor:which_children(?MODULE) ].

init([]) ->
  SupFlags = #{strategy => simple_one_for_one},
  ChildSpec = #{
                id => kv_consumer,
                start => {kv_consumer, start_link, []}
  },
  {ok, {SupFlags, [ChildSpec]}}.

Let's get our supervisor running.
kv_consumer_sup:start_link().
Now, we can start any number of consumers we want. You can even create them with different K values if you'd like. If your queue has any items in it, the consumer will process them immediately once it's started.
kv_consumer_sup:start_child(Db, 1).

  
    
  
  Defining the Producer


Finally, we need a bit of code that will throw items onto the queue with reckless abandon. Good thing we have BEAM.
When you execute this, you should expect the producer to finish before the consumers, assuming you've kept the consumers' K value in the single digits. Either way, the producer should be very quick because there are no key conflicts.
You'll see the log output from each consumer in the meantime, and a "Waiting for push" log when each goes idle.
PushFun = fun(X) ->
  erlfdb:transactional(Db, fun(Tx) -> kv_queue:push(Tx, integer_to_binary(X)) end)
end,

Seq = lists:seq(1, 1000),

timer:tc(fun() ->
           Self = self(),
           [ spawn(fun() -> PushFun(X), Self ! ok end) || X <- Seq ],
           (fun
             Recv(0) -> ok;
             Recv(N) -> receive ok -> Recv(N-1) end
            end)(length(Seq))
        end).

  
    
  
  Cleanup (optional)


To cleanup, you can stop the supervisor and delete the queue.
catch(gen_server:stop(whereis(kv_consumer_sup))),
erlfdb:transactional(Db, fun kv_queue:delete/1).

The primary API for interacting with a FoundationDB database.
Please refer to the FoundationDB C API documentation
for the full specification of the functions we use. In this module documentation,
we won't duplicate all the content from the C API docs. Instead, we'll point out any differences
where relevant, and provide usage examples.
You will usually be interacting with erlfdb functions in the proximity of an
FDB Transaction.
This example shows a basic transaction. See transactional/2 for more.

  
    
  
  Erlang


% Writes to <<"foo">> if it doesn't exist, and keeps track of the number of times we do so.
1> Db = erlfdb:open().
2> erlfdb:transactional(Db, fun(Tx) ->
..   Future = erlfdb:get(Tx, <<"foo">>),
..   case erlfdb:wait(Future) of
..       not_found ->
..           erlfdb:add(Tx, <<"count">>, 1),
..           erlfdb:set(Tx, <<"foo">>, <<"bar">>);
..       _Val ->
..           ok
..   end
.. end).

  
    
  
  Elixir


# Writes to "foo" if it doesn't exist, and keeps track of the number of times we do so.
iex> db = :erlfdb.open()
iex> :erlfdb.transactional(db, fn tx ->
...>
...>   val = tx
...>         |> :erlfdb.get("foo")
...>         |> :erlfdb.wait()
...>
...>   case val do
...>       :not_found ->
...>           :erlfdb.add(tx, "count", 1)
...>           :erlfdb.set(tx, "foo", "bar")
...>       _val ->
...>           :ok
...>   end
...> end).

Advanced floating point number encoding for erlfdb_tuple.

Operations on keys, either by your client Layer, or by the FoundationDB server itself.

The NIF wrapper around all FoundationDB C API function calls.
Tip
Use erlfdb instead.

Creates a database that is to be used as a sandbox.
The sandbox database is constructed using a single

Managing Tenants
Warning
Tenants are currently experimental and are not recommended for use in production.

